

# Evaluation of Gaussian Molecular Integrals

## II. Kinetic-Energy Integrals

Minhhuy Hô  
Julio Manuel Hernández-Pérez

This article carries out the evaluation of kinetic energy integrals using Gaussian-type functions with arbitrary Cartesian angular values. As an example, we calculate the kinetic matrix for the water molecule in the STO-3G basis set.

### ■ Introduction

In this article, the second of a series describing algorithms for evaluating molecular integrals, we detail the evaluation of the kinetic energy integrals. Detailed accounts of molecular integrals can be found in the references of [1]. The electronic kinetic energy in atomic units ( $m_e = e = \hbar = 1/4\pi\epsilon_0 = 1$ ) involves integrals of the type

$$K = \int_{-\infty}^{\infty} \chi_a(\mathbf{r}; \alpha, \mathbf{A}, \mathbf{a}) \left( -\frac{1}{2} \nabla^2 \right) \chi_b(\mathbf{r}; \beta, \mathbf{B}, \mathbf{b}) dr, \quad (1)$$

in which  $\chi_a(\mathbf{r})$  is an unnormalized Cartesian Gaussian *primitive* centered at the nucleus  $\mathbf{A} = \{A_x, A_y, A_z\}$ :

$$\begin{aligned} \chi_a(\mathbf{r}; \alpha, \mathbf{A}, \mathbf{a}) &= (x - A_x)^{a_x} (y - A_y)^{a_y} (z - A_z)^{a_z} e^{-\alpha|\mathbf{r}-\mathbf{A}|^2} = \\ &= (x - A_x)^{a_x} e^{-\alpha|x-A_x|^2} (y - A_y)^{a_y} e^{-\alpha|y-A_y|^2} (z - A_z)^{a_z} e^{-\alpha|z-A_z|^2} = \\ &= \chi_a(x; \alpha, A_x, a_x) \cdot \chi_a(y; \alpha, A_y, a_y) \cdot \chi_a(z; \alpha, A_z, a_z). \end{aligned} \quad (2)$$

Here  $\alpha$  is the orbital exponent and the polynomial represents the angular part, in that the sum of the Cartesian angular momenta  $a_x + a_y + a_z = 0, 1, 2, 3, \dots$  corresponds to functions of type *s*, *p*, *d*, *f*, .... One notable property of the Gaussian function, which will be used here, is that the derivative of a Gaussian function can be expressed as a sum containing Gaussians of lower and higher Cartesian angular values. In particular,

$$\frac{\partial}{\partial x} \chi_a(x; \alpha, A_x, a_x) = a_x \chi_a(x; \alpha, A_x, a_x - 1) - 2\alpha \chi_a(x; \alpha, A_x, a_x + 1). \quad (3)$$

Similarly,  $\frac{\partial}{\partial A_x} \chi_a(x; \alpha, A_x, a_x) = -\frac{\partial}{\partial x} \chi_a(x; \alpha, A_x, a_x)$ . This property is useful in integrals involving differential operators, such as the kinetic energy; in calculations involving the gradient of the energy or Hamiltonian; and in deriving one of the most important algorithms of Gaussian function computation, the recurrence relation.

The kinetic operator, expressed in Cartesian coordinates as  $-\frac{1}{2} \nabla^2 = -\frac{1}{2} \left( \frac{\partial^2}{\partial x^2} + \frac{\partial^2}{\partial y^2} + \frac{\partial^2}{\partial z^2} \right)$ , enables separation of the kinetic integrals into three components  $K = K_x + K_y + K_z$ . For example,

$$K_x = -\frac{1}{2} \int_{-\infty}^{\infty} \chi_a(\mathbf{r}; \alpha, \mathbf{A}, \mathbf{a}) \frac{\partial^2}{\partial x^2} \chi_b(\mathbf{r}; \beta, \mathbf{B}, \mathbf{b}) d\mathbf{r}. \quad (4)$$

Furthermore, if we also separate  $\chi_a$  into its Cartesian components,  $K_x$  becomes a product of three integrals

$$\begin{aligned} K_x = & -\frac{1}{2} \int_{-\infty}^{\infty} \chi_a(x; \alpha, A_x, a_x) \\ & \frac{\partial^2}{\partial x^2} \chi_b(x; \beta, B_x, b_x) dx \int_{-\infty}^{\infty} \chi_a(y; \alpha, A_y, a_y) \cdot \chi_b(y; \beta, B_y, b_y) \\ & dy \int_{-\infty}^{\infty} \chi_a(z; \alpha, A_z, a_z) \cdot \chi_b(z; \beta, B_z, b_z) dz. \end{aligned} \quad (5)$$

## ■ Derivative of Gaussian Functions

The last two integrals in  $K_x$  are simply overlap integrals and can be evaluated as outlined in [1]. Integrating by parts, the first integral is

$$\begin{aligned} & -\frac{1}{2} \left[ \chi_a(x; \alpha, A_x, a_x) \frac{\partial}{\partial x} \chi_b(x; \beta, B_x, b_x) \Big|_{x=-\infty}^{x=\infty} \right. \\ & \quad \left. - \int_{-\infty}^{\infty} \frac{\partial}{\partial x} \chi_a(x; \alpha, A_x, a_x) \frac{\partial}{\partial x} \chi_b(x; \beta, B_x, b_x) dx \right] = \\ & \frac{1}{2} \int_{-\infty}^{\infty} \frac{\partial}{\partial x} \chi_a(x; \alpha, A_x, a_x) \frac{\partial}{\partial x} \chi_b(x; \beta, B_x, b_x) dx \end{aligned} \quad (6)$$

where, according to the property mentioned above, the first term can be expanded into two Gaussian products, which vanish at the integral limits.

Next, we consider the first term of the integrand. Expressing equation (3) in terms of (1) and rearranging, we get

$$\begin{aligned} \frac{\partial}{\partial x} (x - A_x)^{a_x} e^{-\alpha(x-A_x)^2} = \\ a_x(x - A_x)^{a_x-1} e^{-\alpha(x-A_x)^2} + [-2\alpha(x - A_x)] (x - A_x)^{a_x} e^{-\alpha(x-A_x)^2} = [ \\ a_x(x - A_x)^{a_x-1} - 2\alpha(x - A_x)^{a_x+1}] e^{-\alpha(x-A_x)^2}. \end{aligned} \quad (7)$$

Substituting the result into equation (6), we have

$$\begin{aligned} \frac{1}{2} \int_{-\infty}^{\infty} [a_x(x - A_x)^{a_x-1} - 2\alpha(x - A_x)^{a_x+1}] \\ e^{-\alpha(x-A_x)^2} [b_x(x - B_x)^{b_x-1} - 2\beta(x - B_x)^{b_x+1}] e^{-\beta(x-B_x)^2} dx, \end{aligned} \quad (8)$$

and expanding the polynomial part yields

$$\begin{aligned} \frac{1}{2} \int_{-\infty}^{\infty} [a_x b_x (x - A_x)^{a_x-1} (x - B_x)^{b_x-1} - \\ 2\alpha b_x (x - A_x)^{a_x+1} (x - B_x)^{b_x-1} - 2a_x \beta (x - A_x)^{a_x-1} (x - B_x)^{b_x+1} + \\ 4\alpha \beta (x - A_x)^{a_x+1} (x - B_x)^{b_x+1}] e^{-\alpha(x-A_x)^2} e^{-\beta(x-B_x)^2} dx. \end{aligned} \quad (9)$$

We can now express the first integral of equation (5) in terms of one-dimensional Gaussian primitives:

$$\begin{aligned} \frac{1}{2} [a_x b_x \int_{-\infty}^{\infty} \chi_a(x; \alpha, A_x, a_x - 1) \cdot \chi_b(x; \beta, B_x, b_x - 1) dx - \\ 2\alpha b_x \int_{-\infty}^{\infty} \chi_a(x; \alpha, A_x, a_x + 1) \cdot \chi_b(x; \beta, B_x, b_x - 1) dx - \\ 2a_x \beta \int_{-\infty}^{\infty} \chi_a(x; \alpha, A_x, a_x - 1) \cdot \chi_b(x; \beta, B_x, b_x + 1) dx + \\ 4\alpha \beta \int_{-\infty}^{\infty} \chi_a(x; \alpha, A_x, a_x + 1) \cdot \chi_b(x; \beta, B_x, b_x + 1) dx]. \end{aligned} \quad (10)$$

## ■ Kinetic-Energy Integral

We have now shown that the kinetic-energy integral can be written as a product of overlap integrals. The  $x$  component, for example, is given by

$$\begin{aligned} K_x = \frac{1}{2} [a_x b_x \int_{-\infty}^{\infty} \chi_a(x; \alpha, A_x, a_x - 1) \cdot \chi_b(x; \beta, B_x, b_x - 1) dx - \\ 2\alpha b_x \int_{-\infty}^{\infty} \chi_a(x; \alpha, A_x, a_x + 1) \cdot \chi_b(x; \beta, B_x, b_x - 1) dx - \\ 2a_x \beta \int_{-\infty}^{\infty} \chi_a(x; \alpha, A_x, a_x - 1) \cdot \chi_b(x; \beta, B_x, b_x + 1) dx + \\ \end{aligned} \quad (11)$$

$$4\alpha\beta \int_{-\infty}^{\infty} \chi_a(x; \alpha, A_x, a_x + 1) \cdot \chi_b(x; \beta, B_x, b_x + 1) dx \Big] \\ \int_{-\infty}^{\infty} \chi_a(y; \alpha, A_y, a_y) \cdot \chi_b(y; \beta, B_y, b_y) dy \\ \int_{-\infty}^{\infty} \chi_a(z; \alpha, A_z, a_z) \cdot \chi_b(z; \beta, B_z, b_z) dz.$$

## □ Recurrence Relations

Using the notation of [1], where the overlap integral of two Gaussians is expressed in terms of its orthogonal components,

$$S = \int_{-\infty}^{\infty} \chi_a(x; \alpha, A_x, a_x) \cdot \chi_b(x; \beta, B_x, b_x) dx = \\ E_{AB} \left[ \frac{\pi}{\alpha + \beta} \right]^{3/2} s_x(a_x, b_x) s_y(a_y, b_y) s_z(a_z, b_z), \quad (12)$$

the  $x$  component of the kinetic-energy integral, equation (11), can be written as

$$K_x = \frac{1}{2} E_{AB} \left[ \frac{\pi}{\alpha + \beta} \right]^{3/2} \{ a_x b_x s_x(a_x - 1, b_x - 1) - 2\alpha b_x s_x(a_x + 1, b_x - 1) - \\ 2a_x \beta s_x(a_x - 1, b_x + 1) + 4\alpha\beta s_x(a_x + 1, b_x + 1) \} s_y(a_y, b_y) \\ s_z(a_z, b_z) = E_{AB} \left[ \frac{\pi}{\alpha + \beta} \right]^{3/2} k_x(a_x, b_x) s_y(a_y, b_y) s_z(a_z, b_z), \quad (13)$$

where the factor  $\frac{1}{2}$  has been absorbed into the definition of the kinetic integral  $k_x(a_x, b_x)$ . We will derive the recurrence relation for this term  $k_x(a_x, b_x)$ . For  $a_x = b_x = 0$ , the first three terms of equation (13) are zero, and we are left with  $k_x(0, 0) = 2\alpha\beta s_x(1, 1)$ . Analogously, for  $a_x = 0$  or  $b_x = 0$ , we have

$$k_x(a_x, 0) = -a_x \beta s_x(a_x - 1, 1) + 2\alpha\beta s_x(a_x + 1, 1), \\ k_x(0, b_x) = -\alpha b_x s_x(1, b_x - 1) + 2\alpha\beta s_x(1, b_x + 1), \quad (14)$$

and the recurrence relation for the kinetic integral function with any two Cartesian angular momenta generalizes to

$$k_x(a_x, b_x) = \frac{1}{2} [ a_x b_x s_x(a_x - 1, b_x - 1) - 2\alpha b_x s_x(a_x + 1, b_x - 1) - \\ 2a_x \beta s_x(a_x - 1, b_x + 1) + 4\alpha\beta s_x(a_x + 1, b_x + 1) ]. \quad (15)$$

## □ Implementation

The function `Kin` evaluates the kinetic integral of two Gaussian primitives; here `alpha`, `beta`, `RA`, `RB`, `LA`, and `LB` are  $\alpha$ ,  $\beta$ , **A**, **B**, **a**, and **b** as defined earlier. The first step is the evaluation of the overlap integral as described in [1].

```

Kin[alpha_, beta_, RA_, RB_, LA_, LB_] :=
Module[{EAB, Kinetic},
Do[
(* Initial Conditions for overlap recurrence
relation *)

s[i, 0, 0] = 1;
s[i, 1, 0] =
- (RA[[i]] - ((alpha * RA[[i]] + beta * RB[[i]])) /
(alpha + beta));

(* Recurrence Index *)

s[i_, a_, 0] :=
- (RA[[i]] - (alpha * RA[[i]] + beta * RB[[i]]) /
(alpha + beta))
* s[i, a - 1, 0] +
((a - 1) / (2 * (alpha + beta))) * s[i, a - 2, 0];

(* Transfer Equation *)

s[i_, a_, b_] := s[i, a + 1, b - 1] +
(RA[[i]] - RB[[i]]) * s[i, a, b - 1];

(* Initial Conditions for kinetic integral function *)

k[i, 0, 0] = 2 * alpha * beta * s[i, 1, 1];
k[i_, a_, 0] := -a * beta * s[i, a - 1, 1] +
2 * alpha * beta * s[i, a + 1, 1];
k[i_, 0, b_] := -alpha * b * s[i, 1, b - 1] +
2 * alpha * beta * s[i, 1, b + 1];

(* Kinetic Energy Integral *)

k[i_, a_, b_] := (a * b * s[i, a - 1, b - 1]
- 2 * a * beta * s[i, a - 1, b + 1]
- 2 * alpha * b * s[i, a + 1, b - 1]
+ 4 * alpha * beta * s[i, a + 1, b + 1]) / 2;

, {i, 1, 3}]; (* Loop over three Cartesian
components *)

(* Exponential Overlap Factor *)

```

```

EAB = Exp[- (alpha * beta / (alpha + beta)) *
  (RA - RB) . (RA - RB)];

(* Kinetic Energy Integral *)

Kinetic = EAB * ((Pi / (alpha + beta)) ^ (3 / 2))
*
(k[1, LA[[1]], LB[[1]]] * s[2, LA[[2]], LB[[2]]] *
  s[3, LA[[3]], LB[[3]]]
  + s[1, LA[[1]], LB[[1]]] * k[2, LA[[2]], LB[[2]]] *
  s[3, LA[[3]], LB[[3]]]
  + s[1, LA[[1]], LB[[1]]] * s[2, LA[[2]], LB[[2]]] *
  k[3, LA[[3]], LB[[3]]])
]
```

We describe in detail the evaluation of the kinetic-energy matrix for the water molecule ( $r_{\text{OH}} = 1.86942$  bohr,  $\angle_{\text{HOH}} = 100.0269^\circ$ , the geometry optimized at the HF/STO-3G level). The molecule lies in the  $y$ - $z$  plane with Cartesian coordinates in atomic units.

```

R = {{0., 1.43233673, -0.96104039},
  {0., -1.43233673, -0.96104039},
  {0., 0., 0.24026010}};
```

In the STO-3G basis set, each atomic orbital is approximated by a sum of three Gaussians; their unnormalized primitive contraction coefficients and orbital exponents (taken from [2]) are as follows.

```

PrimCoeff = {{0.1543289673, 0.5353281423, 0.4446345422},
  {0.1543289673, 0.5353281423, 0.4446345422},
  {0.1543289673, 0.5353281423, 0.4446345422},
  {-0.09996722919, 0.3995128261, 0.7001154689},
  {0.155916275, 0.6076837186, 0.3919573931},
  {0.155916275, 0.6076837186, 0.3919573931},
  {0.155916275, 0.6076837186, 0.3919573931}};

OrbCoeff = {{3.425250914, 0.6239137298, 0.168855404},
  {3.425250914, 0.6239137298, 0.168855404},
  {130.7093214, 23.80886605, 6.443608313},
  {5.033151319, 1.169596125, 0.38038896},
  {5.033151319, 1.169596125, 0.38038896},
  {5.033151319, 1.169596125, 0.38038896},
  {5.033151319, 1.169596125, 0.38038896}};
```

Here are basis function origins and Cartesian angular values of the orbitals, listed in the order  $H_1(1s)$ ,  $H_2(1s)$ ,  $O_3(1s)$ ,  $O_3(2s)$ ,  $O_3(2p_x)$ ,  $O_3(2p_y)$ , and  $O_3(2p_z)$ .

```
FCenter = {R[[1]], R[[2]], R[[3]], R[[3]], R[[3]],
R[[3]], R[[3]]};
CartAng = {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}, {0, 0, 0},
{1, 0, 0}, {0, 1, 0}, {0, 0, 1}};
```

Specifically, for the kinetic-energy integral of the first primitive of the  $1s$  orbital of hydrogen atom 1,  $\chi_{11}(\mathbf{r})$ , and the first primitive of the  $2p_z$  orbital of the oxygen atom,  $\chi_{71}(\mathbf{r})$ ,

$$\int_{-\infty}^{\infty} \chi_{11}(\mathbf{r}) \left( -\frac{1}{2} \nabla^2 \right) \chi_{71}(\mathbf{r}) d\mathbf{r} = \int_{-\infty}^{\infty} \int_{-\infty}^{\infty} \int_{-\infty}^{\infty} e^{-3.425250914 |\mathbf{r} - \mathbf{R}_{H1}|^2} \left( -\frac{1}{2} \nabla^2 \right) (z - 0.24026010) \times e^{-5.033151319 |\mathbf{r} - \mathbf{R}_{O3}|^2} dx dy dz, \quad (16)$$

where the indices  $\chi_{ij}$  indicate primitive  $j$  of basis function  $i$ . The kinetic integral in terms of the kinetic and overlap function is then

$$E_{AB} \left[ \frac{\pi}{\alpha + \beta} \right]^{3/2} \{k_x(0, 0) s_y(0, 0) s_z(0, 1) + s_x(0, 0) k_y(0, 0) s_z(0, 1) + s_x(0, 0) s_y(0, 0) k_z(0, 1)\}. \quad (17)$$

We start with  $\mathbf{a} = \{0, 0, 0\}$  and  $\mathbf{b} = \{0, 0, 1\}$ , and from the first four equations of the module, we use the recurrence scheme to build up the overlap values needed for  $k_z(0, 1)$ . We will need values up to  $s_z(1, 2)$ ,

$$\begin{aligned} s_z(0, 0) &= 1, \\ s_z(1, 0) &= -\left( H1_z - \frac{\alpha H1_z + \beta O_z}{\alpha + \beta} \right) = 0.714831, \\ s_z(2, 0) &= -\left( H1_z - \frac{\alpha H1_z + \beta O_z}{\alpha + \beta} \right) s_z(1, 0) + \frac{2-1}{2(\alpha + \beta)} s_z(0, 0) = 0.570096, \\ s_z(3, 0) &= -\left( H1_z - \frac{\alpha H1_z + \beta O_z}{\alpha + \beta} \right) s_z(2, 0) + \frac{3-1}{2(\alpha + \beta)} s_z(1, 0) = 0.492034, \\ s_z(1, 1) &= s_z(2, 0) + (H1_z - O_z) s_z(1, 0) = -0.28863, \\ s_z(2, 1) &= s_z(3, 0) + (H1_z - O_z) s_z(2, 0) = -0.192823, \\ s_z(1, 2) &= s_z(2, 1) + (H1_z - O_z) s_z(1, 1) = 0.153908, \end{aligned} \quad (18)$$

leading to  $k_z(0, 1) = -\alpha \cdot 1 \cdot s_z(1, 0) + 2 \alpha \cdot \beta \cdot s_z(1, 2) = 2.85821$ .

Similarly, for  $k_x(0, 0)$ , we will need  $s_x(1, 1)$  which, in turn, comes from  $s_x(1, 0)$  and  $s_x(2, 0)$ . We have

$$\begin{aligned} s_x(1, 0) &= 0, \\ s_x(2, 0) &= 0.0591128, \\ s_x(1, 1) &= 0.0591128, \end{aligned}$$

leading to  $k_x(0, 0) = 2 \alpha \beta s_x(1, 1) = 2.03819$ .

For the y component,

$$\begin{aligned} s_y(1, 0) &= -0.852308, \\ s_y(2, 0) &= 0.785542, \\ s_y(1, 1) &= -0.43525, \end{aligned}$$

leading to  $k_y(0, 0) = 15.0073$ .

With  $E_{AB} = 0.000806444$ , the kinetic integral of equation (16) is  $K = 0.00167343$ , which we can also obtain from the module `Kin`.

```
Kin[OrbCoeff[[1, 1]], OrbCoeff[[7, 1]], FCenter[[1]],
FCenter[[7]], CartAng[[1]], CartAng[[7]]]

0.00167343
```

Three Gaussian primitives  $\chi_a(\mathbf{r})$  for each atomic orbital  $\phi_i(\mathbf{r})$  result in nine integrals of the type that we have just evaluated. For example, the  $K_{17}$  element of the kinetic-energy matrix,

$$K_{17} = \sum_{i=1}^3 \sum_{j=1}^3 N_{1i} N_{7j} d_{1i} d_{7j} \int_{-\infty}^{\infty} \chi_{1i}(\mathbf{r}) \left( -\frac{1}{2} \nabla^2 \right) \chi_{7j}(\mathbf{r}) d\mathbf{r}, \quad (19)$$

is  $\int_{-\infty}^{\infty} \phi_1(\mathbf{r}) \left( -\frac{1}{2} \nabla^2 \right) \phi_7(\mathbf{r}) d\mathbf{r}$ , derived via a contraction scheme that requires the following normalization factor.

$$\mathbf{NormCoeff}[\alpha_, \{\mathbf{ax}_-, \mathbf{ay}_-, \mathbf{az}_-\}] := \left( \frac{2 \alpha}{\pi} \right)^{3/4} \frac{(4 \alpha)^{(\mathbf{ax}+\mathbf{ay}+\mathbf{az})/2}}{((2 \mathbf{ax} - 1) !! (2 \mathbf{ay} - 1) !! (2 \mathbf{az} - 1) !!)^{1/2}};$$



This calculates  $K_{17}$ .

```
Sum[
  NormCoeff[OrbCoeff[[1, i]], CartAng[[1]]]
  * NormCoeff[OrbCoeff[[7, j]], CartAng[[7]]]
  * PrimCoeff[[1, i]]
  * PrimCoeff[[7, j]]
  * Kin[OrbCoeff[[1, i]], OrbCoeff[[7, j]],
    FCenter[[1]], FCenter[[7]],
    CartAng[[1]], CartAng[[7]]]
, {i, 3}, {j, 3}]

-0.167203
```

For larger basis sets, one needs only to replace the summation upper limit 3 in equation (19) with the appropriate number of primitives belonging to a particular basis function and an additional summation for the basis functions.

Finally, here is the resulting kinetic energy matrix.

```
Table[Sum[
  NormCoeff[OrbCoeff[[p, i]], CartAng[[p]]]
  * NormCoeff[OrbCoeff[[q, j]], CartAng[[q]]]
  * PrimCoeff[[p, i]]
  * PrimCoeff[[q, j]]
  * Kin[OrbCoeff[[p, i]], OrbCoeff[[q, j]],
    FCenter[[p]], FCenter[[q]],
    CartAng[[p]], CartAng[[q]]]
, {i, 3}, {j, 3}], {p, 1, 7}, {q, 1, 7}] // MatrixForm //
Chop
```

0.760032	0.0083216	-0.00453868	0.113758	0	0.1
0.0083216	0.760032	-0.00453868	0.113758	0	-0.1
-0.00453868	-0.00453868	29.0032	-0.168011	0	
0.113758	0.113758	-0.168011	0.808128	0	
0	0	0	0	2.52873	
0.199359	-0.199359	0	0	0	2.5
-0.167203	-0.167203	0	0	0	

Since the kinetic-energy matrix is symmetrical, we need only to calculate the upper elements. The value of the elements  $K_{11} = K_{22} = 0.760032$  hartree is the electronic kinetic energy of the hydrogen atom described by the STO-3G basis set (compared to the exact value of 0.5 hartree). Analogously, the element  $K_{33}$  is the corresponding kinetic energy of an electron in the  $1s$  orbital of the oxygen atom, and so forth.

## ■ Conclusion

We have provided an introduction to the evaluation of kinetic-energy integrals involving Gaussian-type basis functions both analytically and by use of recurrence relations. The results are sufficiently general so that no modification of the algorithm is needed when larger basis sets with more Gaussian primitives or primitives with larger angular momenta are employed.

## ■ References

- [1] M. Hô and J. M. Hernández-Pérez, "Evaluation of Gaussian Molecular Integrals I," *The Mathematica Journal*, 2012. doi:10.3888/tmj.14-3.
- [2] "Basis Set Exchange." (April 12, 2012) bse.pnl.gov/bse/portal.

M. Hô and J. M. Hernández-Pérez, "Evaluation of Gaussian Molecular Integrals II," *The Mathematica Journal*, 2013. dx.doi.org/doi:10.3888/tmj.15-1.

## About the Authors

Minhhuy Hô received his Ph.D. in theoretical chemistry at Queen's University, Kingston, Ontario, Canada in 1998. He is currently a professor at the Centro de Investigaciones Químicas at the Universidad Autónoma del Estado de Morelos in Cuernavaca, Morelos, México.

Julio-Manuel Hernández-Pérez obtained his Ph.D. at the Universidad Autónoma del Estado de Morelos in 2008. He has been a professor of chemistry at the Facultad de Ciencias Químicas at the Benemérita Universidad Autónoma de Puebla since 2010.

### **Minhhuy Hô**

*Universidad Autónoma del Estado de Morelos  
Centro de Investigaciones Químicas  
Ave. Universidad, No. 1001, Col. Chamilpa  
Cuernavaca, Morelos, Mexico CP 92010  
homh@uaem.mx*

### **Julio-Manuel Hernández-Pérez**

*Benemérita Universidad Autónoma de Puebla  
Facultad de Ciencias Químicas  
Ciudad Universitaria, Col. San Manuel  
Puebla, Puebla, Mexico CP 72570  
julio.hernandez@correo.buap.mx*